

Evaluation of Gaussian Molecular Integrals

I. Overlap Integrals

Minhhuy Hô
Julio Manuel Hernández-Pérez

This article discusses the evaluation of molecular overlap integrals for Gaussian-type functions with arbitrary angular dependence. As an example, we calculate the overlap matrix for the water molecule in the STO-3G basis set.

■ Introduction

Computational quantum chemistry makes extensive use of various integrals (and their derivatives) of the general form [1, 2, 3]

$$\int_{-\infty}^{\infty} \chi_a(\mathbf{r}) \mathbf{O} \chi_b(\mathbf{r}) d\mathbf{r}, \quad (1)$$

where $\chi_a(\mathbf{r})$ is an unnormalized Cartesian Gaussian function centered at $\mathbf{A} = \{A_x, A_y, A_z\}$:

$$\chi_a(\mathbf{r}; \alpha, \mathbf{A}, \mathbf{a}) = (x - A_x)^{a_x} (y - A_y)^{a_y} (z - A_z)^{a_z} e^{-\alpha|\mathbf{r}-\mathbf{A}|^2}, \quad (2)$$

where \mathbf{A} is normally taken at the nucleus, α is the orbital exponent, and the polynomial represents the angular part, in that the sum of the Cartesian angular momenta $a_x + a_y + a_z = 0, 1, 2, \dots$ corresponds to functions of type s, p, d, f, When the operator \mathbf{O} is 1, one simply has the overlap/density integral; otherwise it can be the energy operator for kinetic energy $-\frac{1}{2} \nabla^2$, electron-nuclear attraction $|\mathbf{r} - \mathbf{R}|^{-1}$, or electron-electron repulsion $|\mathbf{r}_i - \mathbf{r}_j|^{-1}$ (which would involve double integrals). Other molecular properties involving external fields (response functions) or transition moments can also be computed from integrals of this form.

□ Gaussian-Type Functions

Gaussian-type functions are not the most natural choice for expanding the wavefunction. Slater-type functions, where the exponent is $-\alpha |\mathbf{r} - \mathbf{A}|$ instead, can describe atomic systems more realistically. However, complications quickly arise in molecular calculations, which has led to the use of Gaussian functions in the overwhelming majority of current computational programs. Gaussian functions possess several desirable computational properties [4, 5, 6]. (Much credit is due to S. F. Boys for the introduction of the Cartesian Gaussian function into computational chemistry and its early development in a series of 12 papers under the general title “Electronic Wavefunctions,” the first being [5].) A given Slater function can be approximated by a linear combination of several Gaussians.

The first useful property is that the product of two Gaussian functions located at \mathbf{A} and \mathbf{B} is another Gaussian located at a point \mathbf{P} somewhere between \mathbf{A} and \mathbf{B} . (The proof of this can be found in [4].) The product of two Gaussians $\chi_a(\mathbf{r}; \alpha, \mathbf{A}, \mathbf{a})$ and $\chi_b(\mathbf{r}; \beta, \mathbf{B}, \mathbf{b})$ is:

$$\begin{aligned} \chi_a(\mathbf{r}; \alpha, \mathbf{A}, \mathbf{a}) \times \chi_b(\mathbf{r}; \beta, \mathbf{B}, \mathbf{b}) = \\ (x - A_x)^{a_x} (y - A_y)^{a_y} (z - A_z)^{a_z} e^{-\alpha|\mathbf{r}-\mathbf{A}|^2} \times (x - B_x)^{b_x} \\ (y - B_y)^{b_y} (z - B_z)^{b_z} e^{-\beta|\mathbf{r}-\mathbf{B}|^2} = E_{\mathbf{AB}} e^{-(\alpha+\beta)|\mathbf{r}-\mathbf{P}|^2}, \end{aligned} \quad (3)$$

$$\text{with } E_{\mathbf{AB}} = e^{-\frac{\alpha\beta}{\alpha+\beta}|\mathbf{A}-\mathbf{B}|^2} \text{ and } \mathbf{P} = \frac{\alpha\mathbf{A} + \beta\mathbf{B}}{\alpha + \beta}.$$

A second desirable property is that a derivative of a Gaussian can be expressed as a sum of Gaussians of lower and higher Cartesian angular values.

■ Overlap Integrals

The simplest molecular integral is the overlap integral $S = \int_{-\infty}^{\infty} \chi_a(\mathbf{r}; \alpha, \mathbf{A}, \mathbf{a}) \times \chi_b(\mathbf{r}; \beta, \mathbf{B}, \mathbf{b}) d\mathbf{r}$. We first separate the integral into its orthogonal components:

$$S = E_{\mathbf{AB}} S_x(a_x, b_x) S_y(a_y, b_y) S_z(a_z, b_z),$$

where the notation $S_x(a_x, b_x)$ expresses its functional dependence on the Cartesian angular components. The x component, for instance, is:

$$S_x(a_x, b_x) = \int_{-\infty}^{\infty} (x - A_x)^{a_x} (x - B_x)^{b_x} e^{-(\alpha+\beta)(x-P_x)^2} dx. \quad (4)$$

Using a binomial expansion in the polynomial part,

$$(x - A_x)^{a_x} (x - B_x)^{b_x} = \sum_{i_x=0}^{a_x} \sum_{j_x=0}^{b_x} \binom{a_x}{i_x} \binom{b_x}{j_x} (x - P_x)^{i_x+j_x} (P_x - A_x)^{a_x-i_x} (P_x - B_x)^{b_x-j_x}$$

yields:

$$S_x(a_x, b_x) = \sum_{i_x=0}^{a_x} \sum_{j_x=0}^{b_x} \binom{a_x}{i_x} \binom{b_x}{j_x} (P_x - A_x)^{a_x-i_x} (P_x - B_x)^{b_x-j_x} \int_{-\infty}^{\infty} (x - P_x)^{i_x+j_x} e^{-(\alpha+\beta)(x-P_x)^2} dx. \quad (5)$$

Odd values of $i_x + j_x$ result in odd functions whose integrals vanish. For even values of $i_x + j_x$, a solution [7] for the integrals is given by $\left(\frac{\pi}{\alpha+\beta}\right)^{1/2} \frac{(i_x+j_x-1)!!}{[2(\alpha+\beta)]^{(i_x+j_x)/2}}$, and, in those cases:

$$S_x(a_x, b_x) = \sqrt{\frac{\pi}{\alpha+\beta}} \sum_{i_x=0}^{a_x} \sum_{j_x=0}^{b_x} \binom{a_x}{i_x} \binom{b_x}{j_x} \frac{(i_x+j_x-1)!!}{[2(\alpha+\beta)]^{(i_x+j_x)/2}} (P_x - A_x)^{a_x-i_x} (P_x - B_x)^{b_x-j_x}. \quad (6)$$

We keep in mind that for the summations only terms of even values of $i_x + j_x$ survive. We have thus obtained an expression to evaluate the overlap of two Gaussians with arbitrary Cartesian angular factors. Equation (6) should be sufficient for programming or even for manual evaluation with small Cartesian angular values. One can, however, further reduce the number of operations by using a recurrence relation, one of the most useful techniques of computational chemistry. Recurrence relations let us efficiently calculate molecular integrations of higher angular values using previously obtained results with lower angular values. Recurrence relations are used in most computational chemistry programs and their application to other molecular integrations will be shown here.

□ Recurrence Relations

The derivation of the following relations involves straightforward algebraic manipulations, but is rather lengthy. Its omission does not impede our understanding of the recurrence relation, but helps maintain a reasonable continuity in our discussion. Interested readers can find a detailed derivation in the Appendix.

We start by defining the function

$$s_x(a_x, b_x) = \sqrt{\frac{\alpha+\beta}{\pi}} S_x(a_x, b_x),$$

where we have removed the factor $\sqrt{\pi/(\alpha+\beta)}$ from $S_x(a_x, b_x)$ in (6). If we take the derivative of $s_x(a_x, b_x)$ with respect to the nuclear coordinate A_x using the definition of $S_x(a_x, b_x)$ in equation (4),

$$\begin{aligned} \frac{\partial}{\partial A_x} s_x(a_x, b_x) &= \sqrt{\frac{\alpha+\beta}{\pi}} \int_{-\infty}^{\infty} (x-B_x)^{b_x} \frac{\partial}{\partial A_x} (x-A_x)^{a_x} e^{-(\alpha+\beta)(x-P_x)^2} dx = \\ &-a_x s_x(a_x-1, b_x) + 2\alpha s_x(a_x+1, b_x) + 2\alpha(A_x-P_x) s_x(a_x, b_x), \end{aligned} \quad (7)$$

and combining the result with the same derivative of $s_x(a_x, b_x)$, but using the definition of $S_x(a_x, b_x)$ in equation (6) instead,

$$\begin{aligned} \frac{\partial}{\partial A_x} s_x(a_x, b_x) &= \\ &\sum_{i_x=0}^{a_x} \sum_{j_x=0}^{b_x} \binom{a_x}{i_x} \binom{b_x}{j_x} \frac{(i_x+j_x-1)!!}{[2(\alpha+\beta)]^{(i_x+j_x)/2}} \frac{\partial}{\partial A_x} [(P_x-A_x)^{a_x-i_x} (P_x-B_x)^{b_x-j_x}] = \\ &a_x \left(\frac{\alpha}{\alpha+\beta} + 1 \right) s_x(a_x-1, b_x) + b_x \left(\frac{\alpha}{\alpha+\beta} \right) s_x(a_x, b_x-1), \end{aligned} \quad (8)$$

we obtain

$$\begin{aligned} s_x(a_x+1, b_x) &= \\ &-(A_x-P_x) s_x(a_x, b_x) + \frac{a_x}{2(\alpha+\beta)} s_x(a_x-1, b_x) + \frac{b_x}{2(\alpha+\beta)} s_x(a_x, b_x-1). \end{aligned} \quad (9)$$

The same approach in which we combine the derivatives of (4) and (6) with respect to the nuclear coordinate B_x gives

$$\begin{aligned} s_x(a_x, b_x+1) &= \\ &-(B_x-P_x) s_x(a_x, b_x) + \frac{b_x}{2(\alpha+\beta)} s_x(a_x, b_x-1) + \frac{a_x}{2(\alpha+\beta)} s_x(a_x-1, b_x). \end{aligned} \quad (10)$$

For $b_x = 0$, we have the index recursion relation

$$s_x(a_x+1, 0) = -(A_x-P_x) s_x(a_x, 0) + \frac{a_x}{2(\alpha+\beta)} s_x(a_x-1, 0), \quad (11)$$

and combining equations (7) and (8) yields the transfer equation

$$s_x(a_x, b_x+1) = s_x(a_x+1, b_x) + (A_x-B_x) s_x(a_x, b_x). \quad (12)$$

Starting with the initial values $s_x(0, 0) = 1$ and $s_x(1, 0) = -(A_x-P_x)$, the recurrence relation and the transfer equation, we can build up the overlap of functions of higher Cartesian angular values from lower ones. This is particularly useful for contracted Gaussian basis primitives with different functions sharing the same exponent.

□ Implementation

The function `Ov` evaluates the overlap integral of two Gaussian functions; here `alpha`, `beta`, `RA`, `RB`, `LA`, and `LB` are α , β , **A**, **B**, **a**, and **b** as defined earlier.

```

Ov[alpha_, beta_, RA_, RB_, LA_, LB_] :=
Module[{EAB, Overlap},
Do[
(* Initial Conditions *)

s[i_, 0, 0] := 1;
s[i_, 1, 0] :=
- (RA[[i]] - ((alpha * RA[[i]] + beta * RB[[i]]) /
(alpha + beta)));

(* Recurrence Index *)

s[i_, a_, 0] :=
- (RA[[i]] - (alpha * RA[[i]] + beta * RB[[i]]) /
(alpha + beta)) * s[i, a - 1, 0] +
((a - 1) / (2 * (alpha + beta))) * s[i, a - 2, 0];

(* Transfer Equation *)

s[i_, a_, b_] :=
s[i, a + 1, b - 1] + (RA[[i]] - RB[[i]]) * s[i, a, b - 1]

, {i, 1, 3}];

EAB = Exp[- (alpha * beta / (alpha + beta)) *
(RA - RB) . (RA - RB)];

Overlap = EAB * (Pi / (alpha + beta)) ^ (3 / 2) *
s[1, LA[[1]], LB[[1]]] *
s[2, LA[[2]], LB[[2]]] *
s[3, LA[[3]], LB[[3]]]
];

```

■ Examples

The published contracted Gaussian basis sets (see, for example, [8]) are usually not normalized; in our first example, we will calculate the normalization factor of the s and p Cartesian Gaussian functions that we will need later on. The normalization factor is simply the inverse square root of the overlap integral. Here we calculate the overlap between two s functions with the Cartesian angular vector $\mathbf{a} = \{0, 0, 0\}$,

$$S_s = E_{AA} \left(\frac{\pi}{\alpha + \alpha} \right)^{3/2} s_x(0, 0) s_y(0, 0) s_z(0, 0) = \left(\frac{\pi}{2\alpha} \right)^{3/2},$$

and observe that $E_{AA} = s_x(0, 0) = s_y(0, 0) = s_z(0, 0) = 1$. The normalization factor for 1s functions is $1/\sqrt{S_s} = (2\alpha/\pi)^{3/4}$. Similarly, the overlap between two p_x functions ($\mathbf{a} = \{1, 0, 0\}$) is

$$S_{p_x} = E_{AA} \left(\frac{\pi}{\alpha + \alpha} \right)^{3/2} s_x(1, 1) s_y(0, 0) s_z(0, 0) = \left(\frac{\pi}{\alpha + \alpha} \right)^{3/2} s_x(2, 0) = \left(\frac{\pi}{\alpha + \alpha} \right)^{3/2} \frac{1}{4\alpha},$$

where we have analogously used the recurrence relations to obtain $s_x(2, 0)$ and afterward, $s_x(1, 1)$. Similar results are obtained for S_{p_y} and S_{p_z} . The normalization factor for the p function is then $(128\alpha^5/\pi^3)^{1/4}$. These results are special cases of the more general formula of the normalization factor

$$N(\alpha, \mathbf{a}) = \left(\frac{2\alpha}{\pi} \right)^{3/4} \frac{(4\alpha)^{(a_x+a_y+a_z)/2}}{((2a_x-1)!!(2a_y-1)!!(2a_z-1)!!)^{1/2}},$$

and we note that this value depends only on the orbital exponent and the Cartesian angular values.

$$\text{NormCoeff}[\alpha_, \{\mathbf{ax}_, \mathbf{ay}_, \mathbf{az}_\}] := \left(\frac{2\alpha}{\pi} \right)^{3/4} \frac{(4\alpha)^{(\mathbf{ax}+\mathbf{ay}+\mathbf{az})/2}}{((2\mathbf{ax}-1)!!(2\mathbf{ay}-1)!!(2\mathbf{az}-1)!!)^{1/2}};$$

In the second example, we will calculate the overlap matrix of the water molecule ($r_{\text{OH}} = 1.86942$ bohr, $\angle_{\text{HOH}} = 100.0269^\circ$, geometry optimized at the HF/STO-3G level). The molecule lies in the y - z plane with Cartesian coordinates in atomic units.

$$\mathbf{R} = \{ \{0., 1.43233673, -0.96104039\}, \\ \{0., -1.43233673, -0.96104039\}, \\ \{0., 0., 0.24026010\} \};$$

In the STO-3G basis set, each atomic orbital is described by a sum of three Gaussians; here are their primitive contraction coefficients and orbital exponents (taken from [8]).

```

PrimCoeff = {{0.1543289673, 0.5353281423, 0.4446345422},
               {0.1543289673, 0.5353281423, 0.4446345422},
               {0.1543289673, 0.5353281423, 0.4446345422},
               {-0.09996722919, 0.3995128261, 0.7001154689},
               {0.155916275, 0.6076837186, 0.3919573931},
               {0.155916275, 0.6076837186, 0.3919573931},
               {0.155916275, 0.6076837186, 0.3919573931}};

OrbCoeff = {{3.425250914, 0.6239137298, 0.168855404},
               {3.425250914, 0.6239137298, 0.168855404},
               {130.7093214, 23.80886605, 6.443608313},
               {5.033151319, 1.169596125, 0.38038896},
               {5.033151319, 1.169596125, 0.38038896},
               {5.033151319, 1.169596125, 0.38038896},
               {5.033151319, 1.169596125, 0.38038896}};

```

Here are the centers and Cartesian angular values of the orbitals, in the following order: H_{1s} , H_{2s} , O_{1s} , O_{2s} , O_{2p_x} , O_{2p_y} , and O_{2p_z} .

```

FCenter = {R[[1]], R[[2]], R[[3]], R[[3]], R[[3]],
              R[[3]], R[[3]]};

CartAng = {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0},
              {1, 0, 0}, {0, 1, 0}, {0, 0, 1}};

```

For instance, the s atomic orbital of the hydrogen atom 1 is described as

$$\begin{aligned}
 \phi_1(\mathbf{r}) &= \sum_{i=1}^3 N(\alpha_{1i}, \mathbf{a}) \times c_{1i} \times \chi_{1i}(\mathbf{r}; \alpha_{1i}, \mathbf{R}_H, \mathbf{a}) = \\
 &1.7944418 \times 0.1543289673 \times e^{-3.425250914 |\mathbf{r} - \mathbf{R}_H|^2} + \\
 &0.50032649 \times 0.5353281423 \times e^{-0.6239137298 |\mathbf{r} - \mathbf{R}_H|^2} + \\
 &0.18773546 \times 0.4446345422 \times e^{-0.168855404 |\mathbf{r} - \mathbf{R}_H|^2}.
 \end{aligned} \tag{13}$$

Similarly, the p_z orbital of the oxygen atom is

$$\begin{aligned}
 \phi_7(\mathbf{r}) &= 10.745833 \times 0.155916275 \times (z - 0.24026010) \times e^{-5.033151319 |\mathbf{r} - \mathbf{R}_O|^2} + \\
 &1.7337441 \times 0.6076837186 \times (z - 0.24026010) \times e^{-1.169596125 |\mathbf{r} - \mathbf{R}_O|^2} + \\
 &0.42581893 \times 0.391957393 \times (z - 0.24026010) \times e^{-0.38038896 |\mathbf{r} - \mathbf{R}_O|^2}.
 \end{aligned} \tag{14}$$

The overlap integral between the two orbitals expands to nine integrals involving the primitives. Using the function `Ov`, for example, the overlap integral between the first two primitives of equations (13) and (14) is

$$\int_{-\infty}^{\infty} \chi_{11}(\mathbf{r}) \times \chi_{71}(\mathbf{r}) d\mathbf{r} = \int_{-\infty}^{\infty} \int_{-\infty}^{\infty} \int_{-\infty}^{\infty} e^{-3.425250914 |\mathbf{r}-\mathbf{R}_H|^2} \times (z - 0.24026010) \times e^{-5.033151319 |\mathbf{r}-\mathbf{R}_O|^2} dx dy dz.$$

```
Ov[OrbCoeff[[1, 1]], OrbCoeff[[7, 1]],  
   FCenter[[1]], FCenter[[7]], CartAng[[1]], CartAng[[7]]]  
  
-0.0000888019
```

And the overlap between $\phi_1(\mathbf{r})$ and $\phi_7(\mathbf{r})$ is

$$S_{17} = \int_{-\infty}^{\infty} \phi_1(\mathbf{r}) \times \phi_7(\mathbf{r}) d\mathbf{r}.$$

```
Sum[  
  NormCoeff[OrbCoeff[[1, i]], CartAng[[1]]] *  
  NormCoeff[OrbCoeff[[7, j]], CartAng[[7]]] *  
  PrimCoeff[[1, i]] *  
  PrimCoeff[[7, j]] *  
  Ov[  
    OrbCoeff[[1, i]], OrbCoeff[[7, j]],  
    FCenter[[1]], FCenter[[7]],  
    CartAng[[1]], CartAng[[7]]  
  ]  
, {i, 3}, {j, 3}]  
  
-0.245538
```

The overlap matrix **S** for the entire water molecule in the STO-3G basis set can be calculated in a similar manner.


```

Table[Sum[
  NormCoeff[OrbCoeff[[p, i]], CartAng[[p]]] *
  NormCoeff[OrbCoeff[[q, j]], CartAng[[q]]] *
  PrimCoeff[[p, i]] *
  PrimCoeff[[q, j]] *
  Ov[
    OrbCoeff[[p, i], OrbCoeff[[q, j]],
    FCenter[[p]], FCenter[[q]],
    CartAng[[p]], CartAng[[q]]
  ]
, {i, 3}, {j, 3}], {p, 7}, {q, 7}] // MatrixForm //
Chop

```

$$\begin{pmatrix}
1. & 0.250987 & 0.0500166 & 0.454011 & 0 & 0.292761 & -0.245538 \\
0.250987 & 1. & 0.0500166 & 0.454011 & 0 & -0.292761 & -0.245538 \\
0.0500166 & 0.0500166 & 1. & 0.236704 & 0 & 0 & 0 \\
0.454011 & 0.454011 & 0.236704 & 1. & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 1. & 0 & 0 \\
0.292761 & -0.292761 & 0 & 0 & 0 & 1. & 0 \\
-0.245538 & -0.245538 & 0 & 0 & 0 & 0 & 1.
\end{pmatrix}$$

Since the overlap matrix is symmetrical, we need to calculate only the elements above the main diagonal. The basis functions are normalized, as indicated by the unit diagonal elements. We note that S_{15} equals zero. This is the overlap between the 1s orbital of the first hydrogen atom and the $2p_x$ of the oxygen atom. The molecule lies in the y - z plane, so this overlap vanishes due to symmetry. Similar cases occur in the overlaps between the s and p orbitals of O ($S_{35}, S_{36}, S_{37}, \dots$) and among the 2p orbitals (S_{56}, S_{57}, S_{67}), etc. The strongest overlaps are those between the 1s hydrogen orbitals and the 2s oxygen orbital.

We next plot the atomic-orbital overlap between 1s of H_1 and $2p_z$ of O,

$$\sum_{m=1}^3 \sum_{n=1}^3 c_{1m} c_{7n} \chi_m(\mathbf{r}; \alpha_m, \mathbf{R}_H, \mathbf{a}_m) \times \chi_n(\mathbf{r}; \beta_n, \mathbf{R}_O, \mathbf{b}_n),$$

in the y - z plane, superimposing the molecule structure.

```

spz = Sum[PrimCoeff[[1, i]] PrimCoeff[[3, j]]
  Exp[-OrbCoeff[[1, i]]
    ({y, z} - {R[[1, 2]], R[[1, 3]])} .
    ({y, z} - {R[[1, 2]], R[[1, 3]])}]
  (z - R[[3, 3]]) Exp[-OrbCoeff[[1, j]]
    ({y, z} - {R[[3, 2]], R[[3, 3]])} .
    ({y, z} - {R[[3, 2]], R[[3, 3]])}]
, {i, 3}, {j, 3}];

spzFig = ContourPlot[spz, {y, -3, 3}, {z, -3, 3},
  Contours -> 25, PlotPoints -> 100, PlotRange -> All];


```

```

Molecule =
Graphics[{Thickness[0.02], Opacity[0.3], Pink,
  Line[{{R[[1, 2]], R[[1, 3]]}, {R[[3, 2]], R[[3, 3]]}],
  Line[{{R[[2, 2]], R[[2, 3]]}, {R[[3, 2]], R[[3, 3]]}]}];

Show[spzFig, Molecule]

```



We observe a strong distortion of the positive (lower) lobe of the $2p_z$ function toward the hydrogen atom. The negative lobe shows less deformation, whereas the node remains precisely at the atomic position. Note that this is simply the orbital overlap between noninteracting atoms, such as in the case of the “promolecule.”

■ Conclusion

We have provided an introduction to the evaluation of molecular integrals involving Gaussian-type basis functions both analytically and by use of recurrence relations. The results are general and relatively straightforward; the simple algorithm makes it suitable for implementation in a number of programming languages. Together with the kinetic, nuclear-electron attraction, and electron-electron repulsion energies, this is the first step toward the calculation of molecular energies and electronic properties.

■ Appendix

Here we provide the derivation of equations (7) and (8). Differentiate $s_x(a_x, b_x)$ using equation (5) with respect to A_x :

$$\frac{\partial}{\partial A_x} s_x(a_x, b_x) = \sqrt{\frac{\alpha + \beta}{\pi}} \int_{-\infty}^{\infty} (x - B_x)^{b_x} \frac{\partial}{\partial A_x} (x - A_x)^{a_x} e^{-(\alpha + \beta)(x - P_x)^2} dx.$$

Consider the derivative term

$$\begin{aligned} \frac{\partial}{\partial A_x} (x - A_x)^{a_x} e^{-(\alpha + \beta)(x - P_x)^2} = \\ e^{-(\alpha + \beta)(x - P_x)^2} \frac{\partial}{\partial A_x} (x - A_x)^{a_x} + (x - A_x)^{a_x} e^{-(\alpha + \beta)(x - P_x)^2} \frac{\partial}{\partial A_x} [-(\alpha + \beta)(x - P_x)^2]; \end{aligned}$$

the first term is simply

$$-(a_x)(x - A_x)^{a_x - 1} e^{-(\alpha + \beta)(x - P_x)^2},$$

and, using the chain rule, the derivative of the second term is

$$\begin{aligned} -(\alpha + \beta) \frac{\partial}{\partial A_x} [(x - P_x)^2] = \\ (\alpha + \beta) 2(x - P_x) \frac{\partial}{\partial A_x} \left[\frac{\alpha A_x + \beta B_x}{\alpha + \beta} \right] = (\alpha + \beta) 2(x - P_x) \frac{\alpha}{\alpha + \beta} = 2\alpha(x - P_x), \end{aligned}$$

recalling that $P_x = (\alpha A_x + \beta B_x) / (\alpha + \beta)$. Substitute the results into the expression for $\partial s_x(a_x, b_x) / \partial A_x$,

$$\begin{aligned} \frac{\partial}{\partial A_x} s_x(a_x, b_x) = \sqrt{\frac{\alpha + \beta}{\pi}} \int_{-\infty}^{\infty} (x - B_x)^{b_x} [\\ -(a_x)(x - A_x)^{a_x - 1} e^{-(\alpha + \beta)(x - P_x)^2} + 2\alpha(x - P_x)(x - A_x)^{a_x} e^{-(\alpha + \beta)(x - P_x)^2}] dx. \end{aligned}$$

Writing $x - P_x = (x - A_x) + (A_x - P_x)$ in the second term inside the bracket, after expanding we have

$$\begin{aligned} \frac{\partial}{\partial A_x} s_x(a_x, b_x) &= -a_x \sqrt{\frac{\alpha + \beta}{\pi}} \int_{-\infty}^{\infty} (x - A_x)^{a_x-1} (x - B_x)^{b_x} e^{-(\alpha+\beta)(x-P_x)^2} dx + \\ &2\alpha(A_x - P_x) \sqrt{\frac{\alpha + \beta}{\pi}} \int_{-\infty}^{\infty} (x - A_x)^{a_x} (x - B_x)^{b_x} e^{-(\alpha+\beta)(x-P_x)^2} dx + \\ &2\alpha \sqrt{\frac{\alpha + \beta}{\pi}} \int_{-\infty}^{\infty} (x - A_x)^{a_x+1} (x - B_x)^{b_x} e^{-(\alpha+\beta)(x-P_x)^2} dx. \end{aligned}$$

Comparing the three integrals with the definitions of $s_x(a_x, b_x)$ and $S_x(a_x, b_x)$ we have the desired equation (7).

To derive equation (8), we differentiate $s_x(a_x, b_x)$ with respect to A_x using equation (6) instead,

$$\begin{aligned} &\frac{\partial}{\partial A_x} s_x(a_x, b_x) \\ &= \sum_{i_x=0}^{a_x} \sum_{j_x=0}^{b_x} \binom{a_x}{i_x} \binom{b_x}{j_x} \frac{(i_x + j_x - 1)!!}{[2(\alpha + \beta)]^{(i_x+j_x)/2}} \frac{\partial}{\partial A_x} [(P_x - A_x)^{a_x-i_x} (P_x - B_x)^{b_x-j_x}] \\ &= \sum_{i_x=0}^{a_x} \sum_{j_x=0}^{b_x} \binom{a_x}{i_x} \binom{b_x}{j_x} \\ &\quad \frac{(i_x + j_x - 1)!!}{[2(\alpha + \beta)]^{(i_x+j_x)/2}} \left[(a_x - i_x) (P_x - A_x)^{a_x-i_x-1} \left(\frac{\alpha}{\alpha + \beta} - 1 \right) (P_x - B_x)^{b_x-j_x} + \right. \\ &\quad \left. (b_x - j_x) (P_x - B_x)^{b_x-j_x-1} \left(\frac{\alpha}{\alpha + \beta} \right) (P_x - A_x)^{a_x-i_x} \right] \\ &= \left(\frac{\alpha}{\alpha + \beta} - 1 \right) \sum_{i_x=0}^{a_x} \sum_{j_x=0}^{b_x} \binom{a_x}{i_x} \binom{b_x}{j_x} \frac{(i_x + j_x - 1)!!}{[2(\alpha + \beta)]^{(i_x+j_x)/2}} \left[\right. \\ &\quad \left. (a_x - i_x) (P_x - A_x)^{a_x-i_x-1} (P_x - B_x)^{b_x-j_x} \right] + \left(\frac{\alpha}{\alpha + \beta} \right) \\ &\quad \sum_{i_x=0}^{a_x} \sum_{j_x=0}^{b_x} \binom{a_x}{i_x} \binom{b_x}{j_x} \frac{(i_x + j_x - 1)!!}{[2(\alpha + \beta)]^{(i_x+j_x)/2}} \left[(b_x - j_x) (P_x - B_x)^{b_x-j_x-1} (P_x - A_x)^{a_x-i_x} \right]. \end{aligned}$$

We make use of the relationship

$$\binom{a_x}{i_x} (a_x - i_x) = \frac{(a_x - 1)! a_x (a_x - i_x)}{i_x! 1 \times 2 \times 3 \times \dots \times (a_x - i_x - 1) (a_x - i_x)} = a_x \binom{a_x - 1}{i_x}$$

to reduce the derivative above to

$$\begin{aligned} & \frac{\partial}{\partial A_x} s_x(a_x, b_x) \\ &= a_x \left(\frac{\alpha}{\alpha + \beta} - 1 \right) \\ & \quad \sum_{i_x=0}^{a_x-1} \sum_{j_x=0}^{b_x} \binom{a_x-1}{i_x} \binom{b_x}{j_x} \frac{(i_x + j_x - 1)!!}{[2(\alpha + \beta)]^{(i_x+j_x)/2}} [(P_x - A_x)^{a_x-i_x-1} (P_x - B_x)^{b_x-j_x}] + \\ & \quad b_x \left(\frac{\alpha}{\alpha + \beta} \right) \\ & \quad \sum_{i_x=0}^{a_x} \sum_{j_x=0}^{b_x-1} \binom{a_x}{i_x} \binom{b_x-1}{j_x} \frac{(i_x + j_x - 1)!!}{[2(\alpha + \beta)]^{(i_x+j_x)/2}} [(P_x - B_x)^{b_x-j_x-1} (P_x - A_x)^{a_x-i_x}] \\ &= a_x \left(\frac{\alpha}{\alpha + \beta} - 1 \right) s_x(a_x - 1, b_x) + b_x \left(\frac{\alpha}{\alpha + \beta} \right) s_x(a_x, b_x - 1). \end{aligned}$$

This is equation (8), which is what was needed to prove.

■ References

- [1] A. Szabo and N. Ostlund, *Modern Quantum Chemistry: Introduction to Advanced Electronic Structure Theory*, New York: Dover, 1996.
- [2] T. Helgaker, P. Jørgensen, and J. Olsen, *Molecular Electronic-Structure Theory*, New York: Wiley, 2000.
- [3] F. Jensen, *Introduction to Computational Chemistry*, New York: Wiley, 1999.
- [4] I. Shavitt, "The Gaussian Function in Calculations of Quantum Mechanics," in *Methods in Computational Physics*, Vol. 2, B. Alder, S. Fernbach, and M. Rotenberg, eds., New York: Academic Press, 1963 pp. 1–45.
- [5] S. F. Boys, "A General Method of Calculation for Stationary States of Any Molecular System," *Proceedings of the Royal Society A*, **200**, 1950 pp. 542–554. rspa.royalsocietypublishing.org/content/200/1063/542.abstract.
- [6] H. Taketa, S. Huzinaga, and K. Oohata, "Gaussian-Expansion Methods for Molecular Integrals," *Journal of the Physical Society of Japan*, **21**, 1966 pp. 2313–2324.
- [7] M. Abramowitz and I. A. Stegun, *Handbook of Mathematical Functions with Formulas, Graphs, and Mathematical Tables*, New York: Dover, 1965, Eq. (7.4.4) p. 302.

[8] "Basis Set Exchange." (Sep 13, 2011) bse.pnl.gov/bse/portal.

M. Hô and J. M. Hernández-Pérez, "Evaluation of Gaussian Molecular Integrals," *The Mathematica Journal*, 2012. dx.doi.org/doi:10.3888/tmj.14-3.

About the Authors

Minhhuy Hô received his Ph.D. in theoretical chemistry at Queen's University, Kingston, Ontario, Canada in 1998. He is currently a professor at the Centro de Investigaciones Químicas at the Universidad Autónoma del Estado de Morelos in Cuernavaca, Morelos, México.

Julio-Manuel Hernández-Pérez obtained his Ph.D. at the Universidad Autónoma del Estado de Morelos in 2008. He has been a professor of chemistry at the Facultad de Ciencias Químicas at the Benemérita Universidad Autónoma de Puebla since 2010.

Minhhuy Hô

*Universidad Autónoma del Estado de Morelos
Centro de Investigaciones Químicas
Ave. Universidad, No. 1001, Col. Chamilpa
Cuernavaca, Morelos, Mexico CP 92010
homh@uaem.mx*

Julio-Manuel Hernández-Pérez

*Benemérita Universidad Autónoma de Puebla
Facultad de Ciencias Químicas
Ciudad Universitaria, Col. San Manuel
Puebla, Puebla, Mexico CP 72570
julio.hernandez@correo.buap.mx*